Résolution RP-1

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
  with(LinearAlgebra,IdentityMatrix,Determinant,RandomMatrix);
```

$$[IdentityMatrix, Determinant, RandomMatrix] \qquad (1.1)$$

No. 1

a) Avec la syntaxe explicite:

```
> A=Matrix(3,4,[
       [-1,3,6,-2],
       [3,2,-1,3],
       [5,3,-4,5] ]);
```

$$A = \begin{bmatrix} -1 & 3 & 6 & -2 \\ 3 & 2 & -1 & 3 \\ 5 & 3 & -4 & 5 \end{bmatrix} \qquad (2.1)$$

b) Avec la syntaxe implicite:

```
> A=Matrix([
       [-1,3,6,-2],
       [3,2,-1,3],
       [5,3,-4,5] ]);
```

$$A = \begin{bmatrix} -1 & 3 & 6 & -2 \\ 3 & 2 & -1 & 3 \\ 5 & 3 & -4 & 5 \end{bmatrix} \qquad (2.2)$$

c) Comme matrice ligne de vecteurs colonnes:

```
> C1:=<-1,2,5>:
  C2:=<3,2,3>:
  C3:=<6,-1,-4>:
  C4:=<-2,3,5>:
  A=<C1|C2|C3|C4>;
```

$$A = \begin{bmatrix} -1 & 3 & 6 & -2 \\ 2 & 2 & -1 & 3 \\ 5 & 3 & -4 & 5 \end{bmatrix} \qquad (2.3)$$

• Remarque: Une saisie de A sans création explicite des colonnes C_{1j} est bien sûr acceptée

d) Comme matrice colonne de vecteurs lignes:

```
> L1:=<-1|3|6|-2>:
  L2:=<2|2|-1|3>:
  L3:=<5|3|-4|5>:
  A=<L1,L2,L3>;
```

$$A = \begin{bmatrix} -1 & 3 & 6 & -2 \\ 2 & 2 & -1 & 3 \\ 5 & 3 & -4 & 5 \end{bmatrix} \tag{2.4}$$

• Remarque: Une saisie de A sans création explicite des lignes $L_{i\,1}$ est bien sûr acceptée

No. 2

a) Soit la matrice $A_{3\times 5}$ dont les valeurs initiales sont obtenues avec la fonction $f := (i,j) \mapsto (-1)^{i+j} \cdot j^2$

```
> 'A'=Matrix(3,5,(i,j)->(-1)^(i+j)*j^2);
```

$$A = \begin{bmatrix} 1 & -4 & 9 & -16 & 25 \\ -1 & 4 & -9 & 16 & -25 \\ 1 & -4 & 9 & -16 & 25 \end{bmatrix} \tag{3.1}$$

b) Soit la matrice $B_{5\times 5}$ dont les valeurs initiales sont obtenues avec la fonction $f := (i,j) \rightarrow i + 2j$

```
> 'B'=Matrix(5,5,(i,j)->i+2*j);
```

$$B = \begin{bmatrix} 3 & 5 & 7 & 9 & 11 \\ 4 & 6 & 8 & 10 & 12 \\ 5 & 7 & 9 & 11 & 13 \\ 6 & 8 & 10 & 12 & 14 \\ 7 & 9 & 11 & 13 & 15 \end{bmatrix} \tag{3.2}$$

No. 3

a) Soit la matrice A= $\begin{bmatrix} 0 & 1 & -1 \\ 4 & -3 & 4 \\ 3 & -3 & 4 \end{bmatrix}$.

```
> A:=Matrix([
      [0,1,-1],
      [4,-3,4],
      [3,-3,4] ]);
```

$$A := \begin{bmatrix} 0 & 1 & -1 \\ 4 & -3 & 4 \\ 3 & -3 & 4 \end{bmatrix} \tag{4.1}$$

Calculons maintenant la puissance deuxième de A

```
> 'A^2'=A^2;
```

$$A^2 = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \tag{4.2}$$

Ce calcul montre que $A^2 = I$, donc que la matrice A est une matrice involutive.

No. 4

a) Soit la matrice A= $\begin{bmatrix} 7 & 21 & 28 \\ -1 & -3 & -4 \\ -1 & -3 & -4 \end{bmatrix}$

```
> A:=Matrix([
    [7,21,28],
    [-1,-3,-4],
    [-1,-3,-4] ]);
```

$$A := \begin{bmatrix} 7 & 21 & 28 \\ -1 & -3 & -4 \\ -1 & -3 & -4 \end{bmatrix} \tag{5.1}$$

b) Calculons les différentes puissances successives de la matrice A.

```
> A%^2=A^2;
```

$$\begin{bmatrix} 7 & 21 & 28 \\ -1 & -3 & -4 \\ -1 & -3 & -4 \end{bmatrix}^2 = \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \tag{5.2}$$

L'indice de nilpotence de la matrice A est donc 2.

b) Évidemment non car si A est une matrice involutive, on a $A^2 = I$ et donc, pour tout entier $p > 2$, $A^p = I \neq 0$.

No. 5

Générons les deux matrices A et B telles que demandées. (Je suis né le 18 août 1951, donc l'amorce est 510818)

```
> randomize(510818);
```

$$510818 \tag{6.1}$$

```
> A:=RandomMatrix(5,5,generator=-40..60);
  B:=RandomMatrix(5,5,generator=-40..60);
```

$$A := \begin{bmatrix} -9 & -39 & 53 & 15 & -15 \\ -27 & 50 & 20 & 29 & 28 \\ -3 & 14 & 59 & 57 & -28 \\ -1 & -35 & 0 & 28 & 18 \\ 34 & 10 & -6 & 0 & -17 \end{bmatrix}$$

$$B := \begin{bmatrix} 4 & 41 & 12 & 49 & 53 \\ -40 & 1 & 54 & 28 & 60 \\ 39 & -2 & 18 & -37 & -39 \\ -39 & 42 & 49 & 23 & 14 \\ 42 & -28 & -22 & -28 & 27 \end{bmatrix} \qquad \textbf{(6.2)}$$

a)

```
> '(A%+B)'^2=(A+B)^2;
```

$$(A+B)^2 = \begin{bmatrix} 2559 & 636 & 6900 & 3294 & -1941 \\ 3990 & 2170 & 5446 & 542 & -312 \\ -4104 & 2954 & 12013 & 7424 & -2765 \\ 1887 & 646 & 3294 & 524 & -2235 \\ 1698 & -1478 & -200 & 1570 & 2384 \end{bmatrix} \qquad \textbf{(6.3)}$$

```
> '(B%+A)'^2=(B+A)^2;
```

$$(B+A)^2 = \begin{bmatrix} 2559 & 636 & 6900 & 3294 & -1941 \\ 3990 & 2170 & 5446 & 542 & -312 \\ -4104 & 2954 & 12013 & 7424 & -2765 \\ 1887 & 646 & 3294 & 524 & -2235 \\ 1698 & -1478 & -200 & 1570 & 2384 \end{bmatrix} \qquad \textbf{(6.4)}$$

Avec les matrices A et B précédentes, on a bien que $(A+B)^2 = (B+A)^2$.

On arrive aussi à cette conclusion de la manière suivante:

```
> EqualEntries((A+B)^2,(B+A)^2);
```

true **(6.5)**

b)

```
> '(A+B)'^2=(A+B)^2;
```

$$(A+B)^2 = \begin{bmatrix} 2559 & 636 & 6900 & 3294 & -1941 \\ 3990 & 2170 & 5446 & 542 & -312 \\ -4104 & 2954 & 12013 & 7424 & -2765 \\ 1887 & 646 & 3294 & 524 & -2235 \\ 1698 & -1478 & -200 & 1570 & 2384 \end{bmatrix} \qquad \textbf{(6.6)}$$

```
> 'A'^2+2*'A.B'+'B'^2=A^2+2*A.B+B^2;
```

$$A^2+2A\cdot B+B^2=\begin{bmatrix} 4361 & 295 & 5283 & -2740 & -7753 \\ 524 & -479 & 8261 & -3529 & 3039 \\ -4134 & 6369 & 13881 & 7064 & -3195 \\ 3424 & -2400 & 788 & -4435 & -4075 \\ -904 & 2510 & 808 & 6434 & 6322 \end{bmatrix} \tag{6.7}$$

Avec les matrices A et B précédentes, **nous n'avons pas l'égalité** $(A+B)^2=A^2+2AB+B^2$.

```
> EqualEntries((A+B)^2,A^2+2*A.B+B^2);
```

$$false \tag{6.8}$$

c)

```
> '(A+B)'^2=(A+B)^2;
```

$$(A+B)^2=\begin{bmatrix} 2559 & 636 & 6900 & 3294 & -1941 \\ 3990 & 2170 & 5446 & 542 & -312 \\ -4104 & 2954 & 12013 & 7424 & -2765 \\ 1887 & 646 & 3294 & 524 & -2235 \\ 1698 & -1478 & -200 & 1570 & 2384 \end{bmatrix} \tag{6.9}$$

```
> 'A'^2+'A.B'+'B.A'+'B'^2=A^2+A.B+B.A+B^2;
```

$$A^2+A\cdot B+B\cdot A+B^2=\begin{bmatrix} 2559 & 636 & 6900 & 3294 & -1941 \\ 3990 & 2170 & 5446 & 542 & -312 \\ -4104 & 2954 & 12013 & 7424 & -2765 \\ 1887 & 646 & 3294 & 524 & -2235 \\ 1698 & -1478 & -200 & 1570 & 2384 \end{bmatrix} \tag{6.10}$$

Avec les matrices A et B précédentes, **nous avons l'égalité** $(A+B)^2=A^2+AB+BA+B^2$.

```
> EqualEntries((A+B)^2,A^2+A.B+B.A+B^2);
```

$$true \tag{6.11}$$

d)

```
> '(A+B)'^2=(A+B)^2;
```

$$(A+B)^2=\begin{bmatrix} 2559 & 636 & 6900 & 3294 & -1941 \\ 3990 & 2170 & 5446 & 542 & -312 \\ -4104 & 2954 & 12013 & 7424 & -2765 \\ 1887 & 646 & 3294 & 524 & -2235 \\ 1698 & -1478 & -200 & 1570 & 2384 \end{bmatrix} \tag{6.12}$$

```
> '(A+B)'%.'(B%+A)'=(A+B).(B+A);
```

$$(A+B)\cdot(B+A) = \begin{bmatrix} 2559 & 636 & 6900 & 3294 & -1941 \\ 3990 & 2170 & 5446 & 542 & -312 \\ -4104 & 2954 & 12013 & 7424 & -2765 \\ 1887 & 646 & 3294 & 524 & -2235 \\ 1698 & -1478 & -200 & 1570 & 2384 \end{bmatrix} \quad \textbf{(6.13)}$$

Avec les matrices A et B précédentes, **nous avons l'égalité** $(A+B)^2 = (A+B)\,(B+A)$.

```
> EqualEntries((A+B)^2,A^2+A.B+B.A+B^2);
```

$$\mathit{true} \quad \textbf{(6.14)}$$

e)

```
> '(A+B)'^2=(A+B)^2;
```

$$(A+B)^2 = \begin{bmatrix} 2559 & 636 & 6900 & 3294 & -1941 \\ 3990 & 2170 & 5446 & 542 & -312 \\ -4104 & 2954 & 12013 & 7424 & -2765 \\ 1887 & 646 & 3294 & 524 & -2235 \\ 1698 & -1478 & -200 & 1570 & 2384 \end{bmatrix} \quad \textbf{(6.15)}$$

```
> 'A'.'(A%+B)'+'B'.'(A%+B)'=(A+B).(B+A);
```

$$A\cdot(A+B) + B\cdot(A+B) = \begin{bmatrix} 2559 & 636 & 6900 & 3294 & -1941 \\ 3990 & 2170 & 5446 & 542 & -312 \\ -4104 & 2954 & 12013 & 7424 & -2765 \\ 1887 & 646 & 3294 & 524 & -2235 \\ 1698 & -1478 & -200 & 1570 & 2384 \end{bmatrix} \quad \textbf{(6.16)}$$

Avec les matrices A et B précédentes, **nous avons l'égalité** $(A+B)^2 = A\,(A+B) + B\,(A+B)$.

```
> EqualEntries((A+B)^2,(A+B).(B+A));
```

$$\mathit{true} \quad \textbf{(6.17)}$$

f)

```
> Id:=IdentityMatrix(5);
```

$$Id := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(6.18)}$$

```
> 'Id'^2%-'A'^2=Id^2-A^2;
```

(6.19)

$$Id^2 - A^2 = \begin{bmatrix} -449 & 1532 & -1960 & -2175 & 1916 \\ 244 & -3097 & -581 & -2997 & -1291 \\ 1537 & 632 & -3769 & -5320 & -287 \\ -1538 & 2511 & 861 & 247 & 767 \\ 1136 & 1080 & -1750 & -458 & -226 \end{bmatrix} \quad \textbf{(6.19)}$$

```
> '(I-A)'.'(I+A)'=(Id-A).(Id+A);
```

$$(I - A) \cdot (I + A) = \begin{bmatrix} -449 & 1532 & -1960 & -2175 & 1916 \\ 244 & -3097 & -581 & -2997 & -1291 \\ 1537 & 632 & -3769 & -5320 & -287 \\ -1538 & 2511 & 861 & 247 & 767 \\ 1136 & 1080 & -1750 & -458 & -226 \end{bmatrix} \quad \textbf{(6.20)}$$

Avec les matrices A et B précédentes, **nous avons l'égalité** $I^2 - A^2 = (I - A)\,(I + A)$.

```
> EqualEntries(Id^2-A^2,(Id-A).(Id+A));
```

$$true \quad \textbf{(6.21)}$$

g)

```
> 'A.B'^2=(A.B)^2;
```

$$(A \cdot B)^2 = \begin{bmatrix} 8546346 & -10893229 & -3216441 & -17935594 & -18291766 \\ -11302661 & 13427784 & 19374627 & 10426248 & 4876973 \\ -14344067 & 9575004 & 35660548 & -1241168 & 2412452 \\ 3903272 & -5501758 & -3627766 & -6083781 & -4401916 \\ -7170412 & 7373194 & 13432320 & 5427858 & 9299577 \end{bmatrix} \quad \textbf{(6.22)}$$

```
> 'A^2'.'B^2'=(A^2).B^2;
```

$$A^2 \cdot B^2 = \begin{bmatrix} -9627704 & 3962851 & 14295149 & 4092929 & -378988 \\ 13693217 & -10009510 & 6931008 & -21023889 & -7940917 \\ 1504031 & -1401171 & 10551643 & -4934043 & -10041962 \\ -12284443 & 6281936 & 6165891 & 12138940 & 11186199 \\ -849354 & 2957514 & -4499590 & 6300870 & -3104793 \end{bmatrix} \quad \textbf{(6.23)}$$

Avec les matrices A et B précédentes, **nous n'avons pas l'égalité** $AB^2 = A^2\,B^2$.

```
> EqualEntries((A.B)^2,(A^2).B^2);
```

$$false \quad \textbf{(6.24)}$$

h)

```
> 'A^2'-'B^2'=A^2-B^2;
```

$$A^2 - B^2 = \begin{bmatrix} 1291 & -2287 & -1753 & 1632 & -6237 \\ -3578 & 5349 & -17 & 7963 & 3445 \\ -2280 & -1731 & 4041 & 3890 & 613 \\ 1772 & -1430 & -4362 & 2165 & -9 \\ -3792 & -886 & 5120 & -230 & -1514 \end{bmatrix} \quad \textbf{(6.25)}$$

```
> '(A-B)'.'(A+B)'=(A-B).(A+B);
```

$$(A - B) \cdot (A + B) = \begin{bmatrix} 3093 & -2628 & -3370 & -4402 & -12049 \\ -7044 & 2700 & 2798 & 3892 & 6796 \\ -2310 & 1684 & 5909 & 3530 & 183 \\ 3309 & -4476 & -6868 & -2794 & -1849 \\ -6394 & 3102 & 6128 & 4634 & 2424 \end{bmatrix} \quad \textbf{(6.26)}$$

Avec les matrices A et B précédentes, **nous n'avons pas l'égalité** $A^2 - B^2 = (A - B)\,(A + B)$.

```
> EqualEntries(A^2-B^2,(A-B).(A+B));
```

$$\mathit{false} \quad \textbf{(6.27)}$$

No. 6

```
> Test_symétrique:=proc(a::'Matrix'(square))
   local i,j,N_Lignes;
   N_Lignes:=op(1,[op(1,a)]);
   for i from 2 to N_Lignes do
    for j from 1 to i-1 do
     if not evalb(a[i,j]=a[j,i]) then return(`La matrice testée n'est
  pas symétrique`) end if;
    end do
   end do;
   print(`La matrice testée est symétrique`)
  end proc:
```

```
> A:=RandomMatrix(4,5,generator=-20..20);
```

$$A := \begin{bmatrix} 7 & -9 & -16 & -13 & 19 \\ 2 & -4 & -16 & -12 & -12 \\ -13 & -8 & 9 & 16 & 5 \\ -9 & 1 & 5 & -13 & 5 \end{bmatrix} \quad \textbf{(7.1)}$$

```
> 'A'%.'A'^t=A.A^(%T);
```

(7.2)

$$A \cdot A^t = \begin{bmatrix} 916 & 234 & -276 & 112 \\ 234 & 564 & -390 & -6 \\ -276 & -390 & 595 & -29 \\ 112 & -6 & -29 & 301 \end{bmatrix} \quad (7.2)$$

```
> Test_symétrique(rhs((7.2)));
```

La matrice testée est symétrique **(7.3)**

```
> type(A.A^(%T),'Matrix'(symmetric));
```

true **(7.4)**

```
> 'A^t.A'= A^(%T).A;
```

$$A^t \cdot A = \begin{bmatrix} 303 & 24 & -306 & -206 & -1 \\ 24 & 162 & 141 & 24 & -158 \\ -306 & 141 & 618 & 479 & -42 \\ -206 & 24 & 479 & 738 & -88 \\ -1 & -158 & -42 & -88 & 555 \end{bmatrix} \quad (7.5)$$

```
> Test_symétrique(rhs((7.5)));
```

La matrice testée est symétrique **(7.6)**

```
> type(A^(%T).A,'Matrix'(symmetric));
```

true **(7.7)**

No. 7

a)

```
> A:=Matrix([
     [0,r,s],
     [0,0,t],
     [0,0,0] ]);
```

$$A := \begin{bmatrix} 0 & r & s \\ 0 & 0 & t \\ 0 & 0 & 0 \end{bmatrix} \quad (8.1)$$

```
> 'A^3'=A^3;
```

$$A^3 = \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad (8.2)$$

Ce qui vérifie que $A^3 = 0$.

b)

```
> Id:=IdentityMatrix(3);
```

$$Id := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad \textbf{(8.3)}$$

```
> ('Id'-'A')%^(-1)=(Id-A)^(-1);
```

$$(Id - A)^{-1} = \begin{bmatrix} 1 & r & r\,t + s \\ 0 & 1 & t \\ 0 & 0 & 1 \end{bmatrix} \quad \textbf{(8.4)}$$

```
> 'Id%+A%+A^2'=Id+A+A^2;
```

$$Id + A + A^2 = \begin{bmatrix} 1 & r & r\,t + s \\ 0 & 1 & t \\ 0 & 0 & 1 \end{bmatrix} \quad \textbf{(8.5)}$$

Les deux résultats précédents montrent que l'on a bien l'égalité $(I - A)^{-1} = I + A + A^2$.

No. 8

Soit la matrice A= $\begin{bmatrix} k-1 & -1 & 7 \\ 2 & k & 2 \\ 2 & -2 & k+4 \end{bmatrix}$

```
> A:=Matrix([
     [k-1,-1,7],
     [2,k,2],
     [2,-2,k+4] ]);
```

$$A := \begin{bmatrix} k-1 & -1 & 7 \\ 2 & k & 2 \\ 2 & -2 & k+4 \end{bmatrix} \quad \textbf{(9.1)}$$

Calculons son déterminant.

```
> `| A |`=Determinant(A);
```

$$|\,A\,| = k^3 + 3\,k^2 - 12\,k - 28 \quad \textbf{(9.2)}$$

Obtenons la valeurs de k qui annuleront ce déterminant.

```
> Racines:=solve(Determinant(A)=0,{k});
```

$$Racines := \{k = -2\}, \left\{k = -\frac{1}{2} - \frac{\sqrt{57}}{2}\right\}, \left\{k = -\frac{1}{2} + \frac{\sqrt{57}}{2}\right\} \quad \textbf{(9.3)}$$

Donc, pour $k \in \left\{2, -\frac{1}{2} \pm \frac{\sqrt{57}}{2}\right\}$, la matrice A sera une matrice singulière.